
    whisper_ct2

whisper_ct2 is an Elixir library for running OpenAI Whisper speech-to-text
models inside the BEAM. It loads CTranslate2-converted Whisper models through a
Rustler NIF, so Elixir code can transcribe f32 PCM buffers without starting
Python or a separate inference service.
CTranslate2 is the speed-optimised C++ inference engine that powers
faster-whisper — 4-8x faster
than vanilla openai-whisper on the same hardware, with int8 / int8-float16
quantisation and CUDA / oneDNN / MKL / Accelerate backends. 
Installation
def deps do
  [{:whisper_ct2, "~> 0.5"}]
end
Installation downloads a precompiled NIF artefact matching your target triple
from the project's GitHub releases. No Rust toolchain or CMake is needed on
the consumer side.
Source builds
Set WHISPER_CT2_BUILD=1 in your environment (or
config :rustler_precompiled, :force_build, whisper_ct2: true in your parent
project) to compile from source instead. The first source build of CTranslate2
takes ~10 minutes and requires:
	Rust toolchain (rustup, stable)
	cmake, a C++17 compiler, make
	Linux: libstdc++, libgomp available at link time
	CUDA toolkit 12+ if building with cuda or cuda-dynamic features

Models
Point WhisperCt2.load_model/2 at a directory containing a CTranslate2-converted
Whisper model. Required files:
model.bin
config.json
tokenizer.json
vocabulary.txt
preprocessor_config.json
The Systran/faster-whisper-* repositories
ship the first four directly. They do not include
preprocessor_config.json; copy the canonical one from openai/whisper-tiny.en
(or any other openai/whisper-* repo - all Whisper sizes share the same file):
uvx hf download Systran/faster-whisper-tiny.en \
  --local-dir models/faster-whisper-tiny.en

uvx hf download openai/whisper-tiny.en preprocessor_config.json \
  --local-dir models/faster-whisper-tiny.en

Backends
The published Hex package ships four precompiled artefacts; install picks the
right one automatically based on your target triple:
	Target triple	CPU backend	GPU	Notes
	aarch64-apple-darwin	Accelerate	none	Apple Silicon (M1+). Uses Accelerate / AMX paths.
	x86_64-unknown-linux-gnu	oneDNN	cuda-dynamic	Default x86_64 binary; runs well on Intel and AMD.
	x86_64-unknown-linux-gnu (mkl)	Intel MKL	cuda-dynamic	Intel-tuned variant. Opt in via env var (below).
	aarch64-unknown-linux-gnu	oneDNN	cuda-dynamic	Graviton/Grace, optional CUDA on GH200-class hosts.

cuda-dynamic defers loading libcudart until first GPU use, so each artefact
still runs on hosts without CUDA installed. :device selection picks CUDA when
available, otherwise CPU.
x86_64 macOS and Windows are not shipped.
Selecting the MKL variant
For Intel-only fleets where you want maximum SGEMM throughput:
WHISPER_CT2_VARIANT=mkl mix deps.compile whisper_ct2

rustler_precompiled reads this env var at install time and selects the --mkl
artefact instead of the default.
Build from source with a custom backend
For source builds you can pick any combination of ct2rs features:
WHISPER_CT2_BUILD=1 WHISPER_CT2_FEATURES="dnnl cuda-dynamic" mix compile
# other options: mkl, openblas, accelerate, cuda, cuda-dynamic

Runtime device selection
WhisperCt2.available_devices()
#=> {:ok, %{cpu: 1, cuda: 1, cuda_supported: true}}

{:ok, model} =
  WhisperCt2.load_model("models/faster-whisper-tiny.en",
    device: :auto,            # :cpu | :cuda | :auto (default)
    compute_type: :auto,      # :default | :auto | :float16 | :int8_float16 | ...
    device_indices: [0]
  )
:auto picks CUDA when the artefact supports it and at least one CUDA device
is visible; otherwise CPU. Explicit :cuda returns
{:error, %WhisperCt2.Error{reason: :invalid_request}} if either condition
fails.
Usage
{:ok, model} = WhisperCt2.load_model("models/faster-whisper-tiny.en")

# Decode/resample to 16 kHz mono f32 PCM upstream (ffmpeg, Membrane,
# anything that produces little-endian f32 bytes).
pcm = File.read!("jfk.pcm")

{:ok, %WhisperCt2.Transcription{text: text, segments: segs}} =
  WhisperCt2.transcribe(model, {:pcm_f32, pcm}, language: "en")

IO.puts(text)
# => "And so, my fellow Americans ask not what your country can do for you ..."

for s <- segs do
  IO.puts("[#{s.start}-#{s.end}] (no_speech=#{Float.round(s.no_speech_prob, 3)}) #{s.text}")
end
%WhisperCt2.Segment{} carries absolute :start / :end seconds,
:no_speech_prob, :avg_logprob, the underlying text token IDs, and
(when :word_timestamps is on) a list of %WhisperCt2.Word{} with
per-word timing.
Audio contract
CTranslate2 expects mono f32 PCM samples at the model's sample rate
(16 kHz for every published Whisper checkpoint), normalized to the
-1.0..1.0 range. transcribe/3 and transcribe_batch/3 accept exactly
one shape:
	{:pcm_f32, binary} - little-endian f32 samples at the model's
sample rate.

Anything else (paths, raw bare binaries, WAV bytes, MP3, 44.1 kHz, ...)
is rejected at the boundary with an :invalid_request error. There is
no bundled audio decoder; decode, downmix, and resample upstream using
your tool of choice. For a one-shot file conversion:
ffmpeg -i input.mp3 -ar 16000 -ac 1 -f f32le output.pcm

Audio longer than 30 s is chunked into Whisper windows automatically; the
encoder runs once across every chunk in the batch.
Batched transcribe and word timestamps
# Diarization-driven workflow: one master decode upstream, many short
# splices fed in as PCM byte ranges.
samples = File.read!("call.pcm")
turns =
  [
    WhisperCt2.Pcm.slice(samples, 16_000, 0.0, 3.2),
    WhisperCt2.Pcm.slice(samples, 16_000, 3.2, 4.5)
    # ...
  ]
  |> Enum.map(fn {:ok, bin} -> {:pcm_f32, bin} end)

{:ok, transcriptions} =
  WhisperCt2.transcribe_batch(model, turns, language: "en", word_timestamps: true)
transcribe_batch/3 stacks every chunk of every input into one encoder
forward pass. :word_timestamps adds one batched DTW alignment pass and
attaches %Word{} entries to each segment.
Decoding biases
WhisperCt2.transcribe(model, {:pcm_f32, talk_pcm},
  language: "en",
  initial_prompt: "Discussion of CTranslate2, BEAM, and Whisper internals.",
  prefix: "Welcome back to the show."
)
:initial_prompt prepends free-text context (via <|startofprev|>) so the
decoder is biased toward your domain vocabulary or speaker style;
:prefix forces the start of the generated transcript.
Options
transcribe/3 and transcribe_batch/3 accept any subset of:
	Option	Type	Notes
	:language	String.t | nil	ISO code ("en"). nil auto-detects on multilingual.
	:initial_prompt	String.t | nil	Free-text context prepended via <|startofprev|>.
	:prefix	String.t | nil	Forced text the generation must start with.
	:word_timestamps	boolean	Attach per-word timing via a batched DTW alignment.
	:with_timestamps	boolean	Emit <|t_..|> segment timestamps (default true). false for fine-tunes that emit plain text.
	:beam_size	pos_integer	Beam-search width.
	:patience	float	Beam-search patience.
	:length_penalty	float	Decoding length penalty.
	:repetition_penalty	float	Decoding repetition penalty.
	:no_repeat_ngram_size	non_neg_integer	Disallow repeated n-grams of this size.
	:sampling_temperature	float	Sampling temperature.
	:sampling_topk	pos_integer	Top-k sampling.
	:suppress_blank	boolean	Suppress the initial blank token.
	:suppress_tokens	[integer]	Suppress these token IDs.
	:max_length	pos_integer	Max tokens per chunk.
	:num_hypotheses	pos_integer	Number of decoded hypotheses.
	:max_initial_timestamp_index	non_neg_integer	Cap the first timestamp token.

Unset values use the CTranslate2 defaults. no_speech_prob and
avg_logprob are always populated on each segment - there is no opt-in
return-knob.
Unknown option keys and out-of-range values return
{:error, %WhisperCt2.Error{reason: :invalid_request}} before reaching the
NIF.
Errors
All failures return {:error, %WhisperCt2.Error{}}. reason is one of
:invalid_request, :load_error, :inference_error, :runtime_error,
:nif_panic, or :native_error. The struct also implements Exception, so
raise/1 works.
Testing
Unit tests run with no external dependencies:
mix test

The end-to-end transcription test downloads the faster-whisper-tiny.en model
(~75 MB) and the jfk.wav clip from the whisper.cpp samples:
mix test --include integration

Cached under test/fixtures/. Set WHISPER_CT2_REFRESH=1 to redownload.
License
MIT. CTranslate2 itself is MIT-licensed. The bundled ct2rs crate links
CTranslate2 statically by default.


  

    Changelog

0.5.0 - 2026-05-20
Initial public release. Native Elixir Whisper speech-to-text backed by
CTranslate2 through a Rustler NIF over ct2rs::sys::Whisper. No Python.
Features
	WhisperCt2.load_model/2 loads a CTranslate2-converted Whisper model
directory and returns a %WhisperCt2.Model{} with resolved :device
and :compute_type.
	WhisperCt2.transcribe/3 accepts {:pcm_f32, binary} (mono, 16 kHz,
little-endian f32) and returns a %WhisperCt2.Transcription{} whose
:segments carry absolute start/end times, :no_speech_prob,
:avg_logprob, the underlying token IDs, and optional per-word timing.
	WhisperCt2.transcribe_batch/3 stacks every chunk of every input into
one encoder forward pass - a large speedup for diarization-driven
workflows with many short turns.
	:initial_prompt and :prefix bias decoding; :word_timestamps adds a
batched DTW alignment pass attaching %WhisperCt2.Word{} entries;
:with_timestamps toggles <|t_..|> segment timestamps for plain-text
fine-tunes.
	English-only checkpoints (*.en) use the [<|startoftranscript|>]
prompt; multilingual checkpoints use [sot, lang, transcribe].
	WhisperCt2.Pcm.slice/4 carves sub-windows out of an already-decoded
f32 buffer with loud bounds checking.
	WhisperCt2.available_devices/0 reports CPU/CUDA device counts and the
build's CUDA-support flag.
	Structured %WhisperCt2.Error{} taxonomy: :invalid_request,
:load_error, :inference_error, :runtime_error, :nif_panic,
:native_error.

Backends
	Precompiled NIF artefacts via rustler_precompiled for
aarch64-apple-darwin (Accelerate), x86_64-unknown-linux-gnu
(oneDNN, optional mkl variant), and aarch64-unknown-linux-gnu
(oneDNN). CUDA is loaded lazily via cuda-dynamic on every Linux
artefact, so one binary runs on CPU-only and CUDA hosts alike.
	Opt into a source build with WHISPER_CT2_BUILD=1, or pick the MKL
artefact on x86_64 Linux with WHISPER_CT2_VARIANT=mkl.



  

    whisper_ct2 usage rules

Rules for LLM coding agents using whisper_ct2 in a consumer project.
Published per the usage_rules
convention; sync into your project with mix usage_rules.sync.
Load the model once, reuse the struct
WhisperCt2.load_model/2 returns {:ok, %WhisperCt2.Model{ref: ref}} where
ref is a NIF resource pointing at the live CTranslate2 model. The model
stays in memory as long as some process holds the struct. Do not call
load_model/2 per request - hold it in a long-lived process.
defmodule MyApp.Whisper do
  use GenServer

  def start_link(path), do: GenServer.start_link(__MODULE__, path, name: __MODULE__)
  def transcribe(audio, opts \\ []),
    do: GenServer.call(__MODULE__, {:transcribe, audio, opts}, :infinity)

  @impl true
  def init(path) do
    {:ok, model} = WhisperCt2.load_model(path)
    {:ok, model}
  end

  @impl true
  def handle_call({:transcribe, audio, opts}, _from, model) do
    {:reply, WhisperCt2.transcribe(model, audio, opts), model}
  end
end
Put it under your supervision tree. When the process dies the NIF resource
is freed, so let the supervisor reload it.
Parallelism: one Model serialises inside ct2rs
A single %Model{} processes calls serially through the NIF. For real
concurrency across multiple callers, load N replicas (one per process) and
pool them - e.g. with :poolboy, nimble_pool, or a Registry-keyed set
of GenServers. Increasing :max_queued_batches only deepens the queue, not
the worker count.
Do not share the same model across OS threads expecting parallel inference;
share it across BEAM processes for fan-in, not fan-out.
Batched transcribe collapses per-call overhead
WhisperCt2.transcribe_batch(model, [audio1, audio2, ...], opts) stacks
every chunk of every input into one mel batch and runs the encoder once
across the whole thing. For diarization-driven workflows (one call per
turn, dozens to hundreds of turns) this is materially faster than
looping transcribe/3 because CTranslate2 amortises the encoder
forward pass across the batch.
:language applies to every audio in the batch; pass nil to
auto-detect per-audio (only meaningful on multilingual checkpoints).
For carving sub-windows out of an already-decoded buffer, use
WhisperCt2.Pcm.slice(samples, sample_rate, start_s, duration_s) -
it does the f32 byte math (4 bytes/sample) and bounds-checks against the
buffer size, so a slice past the end fails loudly instead of decoding
garbage.
pcm = File.read!("call.pcm")          # f32 LE, 16 kHz mono, prepared upstream
{:ok, turn} = WhisperCt2.Pcm.slice(pcm, model.sampling_rate, 12.3, 4.7)
WhisperCt2.transcribe(model, {:pcm_f32, turn}, language: "en")
Word-level timestamps are opt-in
Pass word_timestamps: true to attach %WhisperCt2.Word{text, start, end, probability} entries to each segment. Implementation reuses the encoder
output from generate and runs one extra batched align call (DTW over
decoder attention) across every chunk in the batch. Cost is on the order
of the alignment pass itself, not a second encoder forward. Use it for
caption alignment or diarization-aware splicing; skip it when you only
need segment timing.
Segment timestamps: :with_timestamps
:with_timestamps defaults to true: the prompt asks Whisper to emit
<|t_..|> tokens that split each 30 s chunk into sub-segments. Leave it
on for stock OpenAI / Systran/faster-whisper-* checkpoints.
Set with_timestamps: false for fine-tunes that ignore the timestamp
instruction or were trained to emit plain text (e.g. some domain
fine-tunes). The chunk's full text then becomes one segment spanning
[0, chunk_duration_s) instead of being silently dropped.
:word_timestamps implicitly forces :with_timestamps back to true -
the DTW alignment needs the timestamp scaffolding. Don't combine
with_timestamps: false with word_timestamps: true and expect the
former to win.
Initial prompt and prefix
	:initial_prompt - free-text conditioning prepended via
<|startofprev|>. Bias the decoder toward domain vocabulary, names,
or speaker style ("Discussion of CTranslate2 internals", "Dialogue
between Alice and Bob"). Same role as in faster-whisper.
	:prefix - forced text the generation must start with. Useful when
the first words are already known (caption corrections, fixed
intro lines).

Both are tokenised inside the NIF without special-token expansion, so
control tokens in the strings are not interpreted.
Pass :language when you know it
:language defaults to nil, which makes Whisper auto-detect from the first
chunk. Auto-detection adds latency and can misfire on short or noisy clips
(English-only fine-tunes still sometimes guess :cy or :fr). Always pass
language: "en" (or the relevant ISO code) when the source language is known.
model.multilingual tells you whether the loaded checkpoint can do anything
other than English - faster-whisper-*.en variants are monolingual and ignore
:language. Branch on model.multilingual if your code supports both.
Result shape
{:ok, %WhisperCt2.Transcription{text, segments, language, duration_s}}:
	text - all segment texts joined by " " and String.trim/1'd. Use this
for display or downstream NLP.
	segments - list of %WhisperCt2.Segment{}, each carrying absolute
:start / :end seconds, :no_speech_prob, :avg_logprob, the
underlying text-token IDs (:tokens), and :words (nil unless
:word_timestamps was set).
	language - resolved ISO code (auto-detected when not pinned).
	duration_s - input audio length in seconds.

Segment timestamps are real fields, not embedded tokens - do not regex
the text for <|t_..|>. Boundaries are produced by Whisper's own timestamp
tokens, parsed inside the NIF.
:no_speech_prob and :avg_logprob are always populated; filter
hallucination with e.g. seg.avg_logprob < -1.0 or
seg.no_speech_prob > 0.6.
Model struct fields are part of the API
Illustrative shape (:ref and :path omitted for brevity; both are
also part of the struct):
%WhisperCt2.Model{
  sampling_rate: 16_000,   # always 16 kHz for published Whisper
  n_samples: 480_000,      # samples in one Whisper window (30 s)
  multilingual: true,      # false for *.en variants
  device: :cpu,            # resolved (never :auto)
  compute_type: :int8,     # resolved (never :default / :auto)
  ...
}
Read these at runtime instead of hardcoding. device and compute_type are
the resolved values - :auto and :default are normalised at load time.
Audio contract is strict
CTranslate2 wants mono f32 PCM at the model's sample rate (always
16 kHz for published Whisper checkpoints), normalised to -1.0..1.0.
transcribe/3 and transcribe_batch/3 accept exactly one shape:
	{:pcm_f32, binary} - little-endian f32 samples at the model's sample
rate.

There is no built-in decoder. Paths, raw bare binaries, WAV bytes,
MP3, etc. are all rejected at the boundary with a clear
:invalid_request error. Decoding, downmixing, and resampling are the
caller's job; use ffmpeg, Membrane, or your platform audio stack
upstream.
ffmpeg -i input.mp3 -ar 16000 -ac 1 -f f32le output.pcm

pcm = File.read!("output.pcm")
WhisperCt2.transcribe(model, {:pcm_f32, pcm}, language: "en")
For microphone or streaming sources, build the f32 buffer yourself and pass
{:pcm_f32, binary}. The authoritative sample rate is model.sampling_rate
on the loaded struct, not a hard-coded 16_000.
Audio longer than 30 s is split into Whisper-window chunks automatically;
per-chunk text is in transcription.segments.
Return shape: never raises on the happy path
Every public function returns {:ok, _} | {:error, %WhisperCt2.Error{}}.
The error struct implements Exception, so raise/1 works if you want
let-it-crash behaviour, but do not write case clauses that assume an
{:ok, _} pattern only. Error.reason is one of:
	:invalid_request - bad options or audio shape; rejected before the NIF
	:load_error - model directory missing or unreadable
	:inference_error - CTranslate2 raised during transcription
	:runtime_error - other ct2rs-side failure
	:nif_panic - Rust panic caught by the panic boundary
	:native_error - fallback for unrecognised native errors

Device and compute_type selection
Probe before deciding:
WhisperCt2.available_devices()
#=> {:ok, %{cpu: 1, cuda: 1, cuda_supported: true}}
	device: :auto (default) picks CUDA when the artefact was built with it
and at least one device is visible; otherwise CPU. Use this unless you
have a reason not to.
	device: :cuda returns
{:error, %WhisperCt2.Error{reason: :invalid_request}} if CUDA is
unavailable - do not assume it succeeds.
	compute_type: :default keeps the stored quantisation of the model
(recommended for Systran/faster-whisper-* int8 builds).
compute_type: :auto lets ct2rs pick the fastest supported on-device.

Do not hardcode :float16 / :int8_float16 unless you know the target
hardware supports it - mismatches raise :load_error.
Model files
load_model/2 needs a directory containing:
model.bin
config.json
tokenizer.json
vocabulary.txt
preprocessor_config.json
Systran/faster-whisper-* ships the first four. preprocessor_config.json
must be copied from any openai/whisper-* repo (all sizes share the file).
A missing preprocessor_config.json is the most common :load_error
cause; check this first when load fails.
Backend selection at install time
The published Hex package picks the right precompiled NIF from your target
triple automatically. Two consumer-facing knobs:
	WHISPER_CT2_VARIANT=mkl on x86_64-unknown-linux-gnu selects the Intel
MKL artefact instead of oneDNN. Only set this on Intel-only fleets.
	WHISPER_CT2_BUILD=1 (or
config :rustler_precompiled, :force_build, whisper_ct2: true) forces a
source build. First build of CTranslate2 takes ~10 minutes and needs
Rust, CMake, and a C++17 toolchain. Do not enable this in CI unless you
understand the cost.

x86_64 macOS and Windows are not shipped - source build only.
Do not
	Do not call load_model/2 per transcription.
	Do not pass .wav (or any other file) paths, raw bare binaries,
encoded WAV bytes, mp3, opus, or non-16 kHz audio to transcribe/3 -
the audio contract is {:pcm_f32, binary} only. Decode and resample
upstream (ffmpeg -ar 16000 -ac 1 -f f32le).
	Do not assume device: :cuda succeeds; check available_devices/0 or
use :auto.
	Do not share a single %Model{} to get parallel inference; pool replicas.
	Do not catch :nif_panic and retry blindly - it indicates a bug worth
reporting.
	Do not hardcode 16_000 as the sample rate - read model.sampling_rate.
	Do not pass :language to a *.en checkpoint and expect anything but
English; check model.multilingual if the language is dynamic.
	Do not regex segment text for <|t_..|> tokens - segment timestamps
are real fields (:start, :end) populated from the model output.
	Do not loop transcribe/3 over a list of short clips when
transcribe_batch/3 would batch them through one encoder pass.
	Do not pass control tokens like <|en|> inside :initial_prompt or
:prefix; they are tokenised as plain text and will not behave as
special tokens.

Native Elixir bindings for Whisper speech-to-text via CTranslate2.
Calls ct2rs::sys::Whisper directly through a Rustler NIF: no Python, no
HTTP gateway. The NIF owns the mel spectrogram, tokenizer, and prompt
construction, so structured per-segment results, :initial_prompt /
:prefix biasing, word-level timestamps, and batched transcribe across
multiple audios are all first-class.
Quickstart
{:ok, model} = WhisperCt2.load_model("/path/to/faster-whisper-tiny")
pcm = File.read!("audio.pcm")

{:ok, %WhisperCt2.Transcription{text: text, segments: segs}} =
  WhisperCt2.transcribe(model, {:pcm_f32, pcm}, language: "en")

IO.puts(text)
for s <- segs, do: IO.puts("[#{s.start}-#{s.end}] #{s.text}")
Audio contract
CTranslate2 expects mono f32 PCM samples at the model's
:sampling_rate (16 kHz for every published Whisper checkpoint),
normalised to the -1.0..1.0 range. transcribe/3 and
transcribe_batch/3 accept exactly one shape:
	{:pcm_f32, binary} - little-endian f32 samples at the model's
sample rate.

Decoding .wav, resampling, downmixing, and any other format
conversion is the caller's job. There is no bundled audio decoder;
use ffmpeg, a dedicated library, or your platform's audio stack
before calling in.
Audio longer than the Whisper 30 s window is chunked internally; the
encoder runs once across every chunk in the batch. Diarization-driven
workflows that need many short splices should use
transcribe_batch/3.

        Audio sources accepted by transcribe/3 and transcribe_batch/3.

        Options accepted by load_model/2.



    


    
      
        transcribe_opt()

      


        Options accepted by transcribe/3 / transcribe_batch/3.



    





  
    Functions
  


    
      
        available_devices()

      


        Reports CTranslate2 device support for this build.



    


    
      
        load_model(path, opts \\ [])

      


        Loads a CTranslate2 Whisper model from a directory.



    


    
      
        transcribe(model, audio, opts \\ [])

      


        Transcribes audio using model.



    


    
      
        transcribe_batch(model, audios, opts \\ [])

      


        Transcribes a list of audios in one batched generate call. Every
chunk of every input shares a single encoder forward pass; output
preserves input order.

          @type audio() :: {:pcm_f32, binary()}


      


Audio sources accepted by transcribe/3 and transcribe_batch/3.

          @type load_opt() ::
  {:device, :cpu | :cuda | :auto}
  | {:compute_type, WhisperCt2.Model.compute_type()}
  | {:device_indices, [non_neg_integer(), ...]}
  | {:num_threads_per_replica, non_neg_integer()}
  | {:max_queued_batches, integer()}
  | {:cpu_core_offset, integer()}


      


Options accepted by load_model/2.

          @type transcribe_opt() ::
  {:language, String.t() | nil}
  | {:initial_prompt, String.t() | nil}
  | {:prefix, String.t() | nil}
  | {:word_timestamps, boolean()}
  | {:with_timestamps, boolean()}
  | {:beam_size, pos_integer()}
  | {:patience, float()}
  | {:length_penalty, float()}
  | {:repetition_penalty, float()}
  | {:no_repeat_ngram_size, non_neg_integer()}
  | {:sampling_temperature, float()}
  | {:sampling_topk, pos_integer()}
  | {:suppress_blank, boolean()}
  | {:max_length, pos_integer()}
  | {:num_hypotheses, pos_integer()}
  | {:max_initial_timestamp_index, non_neg_integer()}
  | {:suppress_tokens, [integer()]}


      


Options accepted by transcribe/3 / transcribe_batch/3.

  


        

      

      
        Functions


        


  
    
      
    
    
      available_devices()



        
          
        

    

  


  

      

          @spec available_devices() ::
  {:ok,
   %{cpu: non_neg_integer(), cuda: non_neg_integer(), cuda_supported: boolean()}}
  | {:error, WhisperCt2.Error.t()}


      


Reports CTranslate2 device support for this build.
Returns {:ok, %{cpu: n, cuda: n, cuda_supported: bool}} on success.
cuda_supported reflects compile-time CUDA features (build with
WHISPER_CT2_FEATURES=cuda-dynamic mix compile to enable). cuda is the
count of NVIDIA GPU devices visible at runtime, or 0 when CUDA is not
built in.

          @spec load_model(Path.t(), [load_opt()]) ::
  {:ok, WhisperCt2.Model.t()} | {:error, WhisperCt2.Error.t()}


      


Loads a CTranslate2 Whisper model from a directory.
See the WhisperCt2 module doc for required model files.
Options
	:device - :cpu, :cuda, or :auto (default). :auto picks CUDA
when the binary was built with CUDA support and a device is visible;
otherwise CPU.
	:compute_type - precision used at inference. :default keeps the
model's stored quantisation; :auto picks the fastest supported on
this device.
	:device_indices - non-empty list of GPU indices (default [0]).
	:num_threads_per_replica - intra-op threads. 0 lets CTranslate2 pick.
	:max_queued_batches, :cpu_core_offset - passed through to
CTranslate2.


  



    

  
    
      
    
    
      transcribe(model, audio, opts \\ [])



        
          
        

    

  


  

      

          @spec transcribe(WhisperCt2.Model.t(), audio(), [transcribe_opt()]) ::
  {:ok, WhisperCt2.Transcription.t()} | {:error, WhisperCt2.Error.t()}


      


Transcribes audio using model.
Returns {:ok, %WhisperCt2.Transcription{}} whose :segments carry
absolute start/end times, no_speech_prob, avg_logprob, the
underlying text tokens, and (when :word_timestamps is set) per-word
timing. no_speech_prob and avg_logprob are always populated.
Options
	:language - ISO code ("en"). nil (default) auto-detects.
	:initial_prompt - free-text context prepended via <|startofprev|>
to bias decoding.
	:prefix - forced text the generation must start with.
	:word_timestamps - when true, attaches :words to each segment
via one extra batched DTW alignment pass. Default false.
	:with_timestamps - when true (default) the prompt asks the model
to emit <|t_..|> timestamp tokens that split the output into
sub-segments. Set to false for fine-tunes that emit text without
timestamps; the chunk's full text becomes one segment spanning
[0, chunk_duration_s). Implicitly forced to true whenever
:word_timestamps is enabled because alignment needs the timestamp
scaffolding.
	Decoding knobs forwarded to CTranslate2: :beam_size, :patience,
:length_penalty, :repetition_penalty, :no_repeat_ngram_size,
:sampling_temperature, :sampling_topk, :suppress_blank,
:max_length, :num_hypotheses, :max_initial_timestamp_index,
:suppress_tokens.


  



    

  
    
      
    
    
      transcribe_batch(model, audios, opts \\ [])



        
          
        

    

  


  

      

          @spec transcribe_batch(WhisperCt2.Model.t(), [audio()], [transcribe_opt()]) ::
  {:ok, [WhisperCt2.Transcription.t()]} | {:error, WhisperCt2.Error.t()}


      


Transcribes a list of audios in one batched generate call. Every
chunk of every input shares a single encoder forward pass; output
preserves input order.
Options are the same as transcribe/3. :language applies to every
audio in the batch; pass nil to auto-detect per-audio.

  


        

      


  

    
WhisperCt2.Model 
    



      
Loaded Whisper model handle.
Holds an opaque NIF reference plus cached metadata (:sampling_rate,
:n_samples, :multilingual). The reference is garbage-collected by the
BEAM when no longer reachable; CTranslate2 frees the model at that point.

        device()

      


    


    
      
        t()

      compute_type()



        
          
        

    

  


  

      

          @type compute_type() ::
  :default
  | :auto
  | :float32
  | :float16
  | :bfloat16
  | :int8
  | :int8_float32
  | :int8_float16
  | :int8_bfloat16
  | :int16

          @type device() :: :cpu | :cuda


      



  



  
    
      
    
    
      t()



        
          
        

    

  


  

      

          @type t() :: %WhisperCt2.Model{
  compute_type: compute_type(),
  device: device(),
  multilingual: boolean(),
  n_samples: pos_integer(),
  path: Path.t(),
  ref: reference(),
  sampling_rate: pos_integer()
}


      



  


        

      


  

    
WhisperCt2.Native 
    



      
Low-level Rustler bindings to ct2rs::sys::Whisper.
This module is private to the library. Use WhisperCt2 for the public API.
Stub names must match the Rust NIF symbols verbatim (Rustler verifies arity
at module load time); user-friendly wrappers live below them.

      


      
        Summary


  
    Functions
  


    
      
        available_devices()

      


        CPU/CUDA device counts and the build's CUDA-support flag.



    


    
      
        load_model(path, opts)

      


        Loads a CT2 Whisper model directory.



    


    
      
        model_info(model)

      


        Returns model metadata.



    


    
      
        transcribe(model, samples_bin, opts)

      


        Runs Whisper on a buffer of PCM samples.



    


    
      
        transcribe_batch(model, samples_bins, opts)

      


        Runs Whisper on a list of PCM sample buffers in one batched generate
call. Returns a list of structured transcription maps in input order.

          @spec available_devices() :: {:ok, map()} | {:error, map()}


      


CPU/CUDA device counts and the build's CUDA-support flag.

  



  
    
      
    
    
      load_model(path, opts)



        
          
        

    

  


  

      

          @spec load_model(String.t(), map()) :: {:ok, reference()} | {:error, map()}


      


Loads a CT2 Whisper model directory.

  



  
    
      
    
    
      model_info(model)



        
          
        

    

  


  

      

          @spec model_info(reference()) :: {:ok, map()} | {:error, map()}


      


Returns model metadata.

  



  
    
      
    
    
      transcribe(model, samples_bin, opts)



        
          
        

    

  


  

      

          @spec transcribe(reference(), binary(), map()) :: {:ok, map()} | {:error, map()}


      


Runs Whisper on a buffer of PCM samples.
samples_bin is a binary of little-endian f32 samples (mono, 16 kHz).
Audio longer than the 30 s Whisper window is chunked and batched
internally; the encoder runs once across all chunks. Returns a structured
transcription map (%{language, duration_s, segments: [...]}).

  



  
    
      
    
    
      transcribe_batch(model, samples_bins, opts)



        
          
        

    

  


  

      

          @spec transcribe_batch(reference(), [binary()], map()) ::
  {:ok, [map()]} | {:error, map()}


      


Runs Whisper on a list of PCM sample buffers in one batched generate
call. Returns a list of structured transcription maps in input order.

  


        

      


  

    
WhisperCt2.Pcm 
    



      
Helpers for slicing little-endian f32 PCM buffers without re-decoding the
source audio.
A diarization-driven workflow typically decodes the master audio to
f32 PCM once (upstream of this library) and then runs many short
transcribe calls over per-turn slices. slice/4 does the byte math
(4 bytes per sample at sample_rate samples/second) and bounds-checks
against the buffer size.

        slice(samples, sample_rate, start_s, duration_s)

      


        Returns the f32-PCM bytes for [start_s, start_s + duration_s).

      slice(samples, sample_rate, start_s, duration_s)



        
          
        

    

  


  

      

          @spec slice(binary(), pos_integer(), number(), number()) ::
  {:ok, binary()} | {:error, WhisperCt2.Error.t()}


      


Returns the f32-PCM bytes for [start_s, start_s + duration_s).
samples is a binary of little-endian f32 samples at sample_rate
samples/second. Both start_s and duration_s are seconds; either
integer or float.
Returns {:error, %WhisperCt2.Error{reason: :invalid_request}} when:
	start_s < 0;
	duration_s is zero or negative, or rounds to zero samples;
	the requested window is not fully contained in the buffer
(start_sample + requested_samples > total_samples).

The range is not clamped — slicing past the end is a caller bug
and we surface it loudly. Bound your duration before calling, or use
min(duration_s, buffer_duration - start_s).

  


        

      


  

    
WhisperCt2.Segment 
    



      
One <|t_start|> text <|t_end|> segment of a transcription.
Times are absolute seconds within the input audio. tokens is the raw
text-token ID list (timestamp tokens stripped); useful for diarization or
custom decoding. no_speech_prob is the no-speech probability of the
parent 30 s chunk, repeated on every segment in that chunk. avg_logprob
is the sequence-level average log probability returned by CTranslate2 -
filter at e.g. avg_logprob < -1.0 to reject low-confidence hallucination.
words is nil unless :word_timestamps was set on the transcribe call;
when present it carries one %WhisperCt2.Word{} per Whisper word with its
own time span.

          @type t() :: %WhisperCt2.Segment{
  avg_logprob: float(),
  end: float(),
  no_speech_prob: float(),
  start: float(),
  text: String.t(),
  tokens: [non_neg_integer()],
  words: [WhisperCt2.Word.t()] | nil
}


      



  


        

      


  

    
WhisperCt2.Transcription 
    



      
Result of a WhisperCt2.transcribe/3 call.
text is the concatenated, whitespace-trimmed transcript across every
segment. segments is the structured per-<|t_..|> decomposition
produced by CTranslate2, with absolute start/end times in seconds,
no_speech_prob, sequence-level avg_logprob, and the underlying token
IDs. language is the resolved ISO code (auto-detected when not pinned).
duration_s is the input audio length, useful for VAD/diarization
pipelines that hand short splices in.

      


      
        Summary


  
    Types
  


    
      
        t()

      


    





      


      
        Types


        


  
    
      
    
    
      t()



        
          
        

    

  


  

      

          @type t() :: %WhisperCt2.Transcription{
  duration_s: float(),
  language: String.t(),
  segments: [WhisperCt2.Segment.t()],
  text: String.t()
}


      



  


        

      


  

    
WhisperCt2.Word 
    



      
One word produced by :word_timestamps.
Times are absolute seconds within the input audio. probability is the
mean of per-token acoustic probabilities from the DTW alignment pass.

          @type t() :: %WhisperCt2.Word{
  end: float(),
  probability: float(),
  start: float(),
  text: String.t()
}


      



  


        

      


  

    
WhisperCt2.Error exception
    



      
Structured error returned from WhisperCt2 calls.
reason is one of:
	:invalid_request - bad arguments (bad path, malformed audio, oversized chunk).
	:load_error      - CTranslate2 could not load the model directory.
	:inference_error - the Whisper model returned an error during generation.
	:runtime_error   - internal NIF runtime fault (e.g. poisoned mutex).
	:nif_panic       - the Rust side panicked; should never happen in practice.
	:native_error    - fallback for unrecognised native error types.

        new(reason, message, details \\ %{})

          @type reason() ::
  :invalid_request
  | :load_error
  | :inference_error
  | :runtime_error
  | :nif_panic
  | :native_error

          @type t() :: %WhisperCt2.Error{
  __exception__: true,
  details: map(),
  message: String.t(),
  reason: reason()
}

          @spec from_native(map()) :: t()

          @spec new(reason(), String.t(), map()) :: t()

